
Abstract 
The paper discusses three general classes of nonlinear programming models. Our objective is to 
show how to represent optimization problem, such as nonlinear programs, in a compact format 
using extended mathematical programming. This is a useful tool, especially in cases when 
complementarity representation, such as mixed complementarity problems, could be difficult to 
apply. We reflect on the special features of the abstract neoclassical growth model with infinite 
horizon and illustrate four distinct approaches to computing equilibrium transition paths.  The 
relationship between optimization and equilibrium modeling is explored. We conclude with a test of 
a new model representation (extended mathematical programming) using two common general 
equilibrium studies in the literature: Ramsey and Negishi. The new framework allows to define 
complementarity problems for a model formulated as an optimization problem, thus the resulting 
model becomes in fact a complementarity problem. 

1 Introduction

The finite-dimensional nonlinear optimization model can be written as:

Given: f : Rn
→ R, g : Rn

→ Rm, ℓ, u ∈ Rn

Find: max
x∈Rn

f(x) (1)

s.t. g(x) = 0, ℓ ≤ x ≤ u

where f(.) is the objective function, g(.) describes the constraints, x is decision variable, ℓ and u
represents lower and upper bounds respectively. The standard analytic technique for finding the
solution to nonlinear optimization problem is the method of Lagrange multipliers, reducing the
optimization to solution of a Karush-Kuhn-Tucker (KKT) system of inequalities. It can be solved
numerically as a nonlinear program (NLP). The nonlinear program is a fundamental building block
of economic equilibrium models, as it is used to describe the decision problems facing most of
economic agents.

Another class of mathematical programs is complementarity model. It can be solved numerically
as a mixed complementarity problem (MCP). This is particularly useful model when computing
market equilibrium which cannot be represented by an optimization model. The MCP has no
objective variable and is defined as:

Given: F : RN
→ RN , ℓ, u ∈ RN

Find: v,w, z ∈ RN (2)

s.t. F (z) = w − v, w ≥ 0, v ≥ 0, ℓ ≤ z ≤ u

wT (z − ℓ) = vT (u− z) = 0

where F (.) is the objective function, v and w are slack variables, z is decision variable. The MCP
format involves the definition of complementarity pairs for variables and equations. This means
that bounded variables must be mapped to complementarity inequality. By virtue of the KKT
conditions, the complementarity problem class is large, including linear and nonlinear
optimization problems as special cases. For cases in which the MCP corresponds to an optimization
problem, the MCP represents the KKT system which has a particular skew-symmetric structure.
The converse is not always the case: while all optimization problems can be expressed as
complementarity problems, not all complementarity problems can be written down as
optimization problems.

A new method, developed by Ferris et al. (2009), for representing complementarity problems
through extended mathematical programming (EMP). This general method is diverse, including
hierarchical models, bilevel optimization, disjunctive programs, and embedded complementarity
problems. In this paper we focus on the last of these problem types which permits specification of
a complementarity problem to be solved simultaneously with an optimization problem. We use
General Algebraic Modeling System (GAMS) to describe this approach in details.

The idea of the EMP approach is to translate optimization problem into complementarity problem.
An optimization model can be expressed as two linked models. These includes:

(i) an optimization problem defined in terms of decision variable, x, taking explicit multipliers,
λ, and auxiliary variables, y, as given;

(ii) a set associated with auxiliary constraints which implicitly determine y.

The combined optimization model can be written as:

Given: f : Rn+M+m
→ R, g : Rn+M+m

→ Rm, H : Rn+M+m
→ RM

Find: min
x|y,λ

f(x; y, λ) (3)

s.t. g(x; y, λ) ≥ 0 ⊥ λ ≥ 0,

where y solves H(x, y, λ) = 0

In this problem, H(.) describes auxiliary constraints.1 The NLP objective function f(.) and
constraints g(.) are written with a semi-colon to separate the decision variable, x, from the
remaining variables, y and λ, which are treated as parametric constants. Please note that y is
treated as a parameter in the optimization problem, but it is a variable in the EMP formulation.

The EMP format requires to specify how to partition variables and equations in the model. That
is, we need to indicate which variables are explicit multipliers2 and which are auxiliary variables.
Further, we need to indicate which equations are auxiliary constraints, i.e. equations that define
H(.) rather than g(.). The complementarity problem corresponding to (3) has the form:

∇xf(x; y, λ)− λT∇xg(x; y, λ) = 0 ⊥ x free

g(x; y, λ) ≥ 0 ⊥ λ ≥ 0 (3a)

H(x, y, λ) = 0 ⊥ y free

This partitioning information is written in GAMS to an information file at execution time. The
file indicates that H(.) is associated with variable y. It would also indicate that variable λ
represents the Lagrange multiplier associated with g(x, y). The solver creates such mixed
complementarity model (3a) consisting of the first order conditions for the nonlinear optimization
problem together with the complementarity conditions which implicitly determine y.

These finite-dimensional programming formats (NLP, MCP and EMP) are useful tools for processing
infinite horizon models. Our paper demonstrates a few ways in which these tools can be applied

1In the GAMS implementation of EMP, auxiliary variables and auxiliary equations are defined in pairs, so there are
always exactly as many auxiliary variables as there are auxiliary constraints. The EMP example shown here assumes
that all auxiliary constraints are equations and auxiliary variables are unconstrained. The auxiliary variable could
alternatively been subject to upper and lower bounds, in which case the auxiliary constraint would be interpreted as
a complementarity condition, i.e.:

H(x, y, λ) ≷ 0 ⊥ ℓ ≤ y ≤ u

2These are dual (primal) variables in a primal (dual) optimization problem.

to the same model, producing identical results. We use a neoclassical dynamic model as an
example of application for EMP, because the model is well studied in the economics literature. The
rest of the paper is organised as following. Section 2 describes a logic of the Ramsey’s growth
model as an infinite horizon NLP. Next we define this model as an MCP and explain calibration
issues. There are few characteristic features of the dynamic model which we discuss at greater
length in section 4. In section 5, the EMP format is applied for this model and for one other CGE
model (welfare model Negishi) in order to show how EMP works. Finally, we make a conclusion
regarding the discussed model classes. The GAMS codes are available in the appendix. We
indicate parameters and exogenous variables by lowercase letters to distinguish from the
endogenous variables (uppercase letters).

2 Ramsey’s Optimization Problem

One special type of complementarity problem (2) is the Arrow-Debreu computable general
equilibrium (CGE) model. A general-equilibrium model consists of (i) profit-maximizing firms, (ii)
markets, typically with supply and demand mediated through prices, and (iii) budget-constrained
utility-maximizing households. Production activities in the Arrow-Debreu framework transform
some goods and factors into others goods. These may include trade activities which transform
domestic into foreign goods, activities which transform leisure into labor supply, and more
conventional production activities which convert labor, capital and material inputs into products.

A CGE model we may represent as an optimization problem (1) and it should produce the same
results as an equilibrium problem (2). To show this we will use a common model in economics -
the neoclassical growth model. It is often presented as a Ramsey infinite-horizon dynamic
optimization problem:

max U(Ct) =

∞
∑

t=0

(

1

1 + ρ

)t C1−θ
t − 1

1− θ
(4)

s.t. Ct = f(Kt)− It, Kt+1 = (1− δ)Kt + It, K0 = K̄0, It ≥ 0

where for simplicity it is assumed that there is no population growth. Ct, Kt, and It are
aggregate consumption, capital, and investment in period t respectively, ρ is the time preference
rate, δ is the rate of depreciation, and θ is an elasticity of the marginal utility of consumption (it
is an inverse of the intertemporal elasticity of substitution σ = 1/θ). It is typically assumed that
the marginal product of capital is positive but decreasing, i.e. f ′(K) > 0 and f ′′(K) < 0. The
initial capital stock in period t = 0 is specified exogenously.

The maximand is a constant-elasticity-of-intertemporal substitution (CEIS) utility function. This
additively separable utility function is simply a monotonic transformation of conventional CES

utility function:

Û(Ct) =

[

∞
∑

t=0

(

1

1 + ρ

)t

C1−θ
t

]1/1−θ

(4a)

The equivalence of U(.) and Û(.) is apparent when recalling that a monotonic transformation of
utility does not alter the underlying preference ordering:

Û = V(U) = [aU + κ]1/a

where κ =

∞
∑

t=0

(

1

1 + ρ

)t

=
1 + ρ

ρ
and a = 1− θ

V(.) is a monotonic transformation (V ′ > 0), hence optimization of U(.) and Û(.) yield identical
demand functions:

Ct =
M

E1−σ

(

ρ

(1 + ρ)t+1Pt

)σ

(5)

where E(p) =

(

ρ

1 + ρ

)σ/(1−σ)
(

∑

t

(

1

1 + ρ

)tσ

P 1−σ
t

)1/(1−σ)

(5a)

where M is income, Pt is market price, and E(.) is expenditure function. Alternatively, recall that
preference orderings are defined by the marginal rate of substitution. In both of these models (4)
and (4a) we have:

∂U/∂Ct+1

∂U/∂Ct
=
∂Û/∂Ct+1

∂Û/∂Ct

=
1

1 + ρ

(

Ct

Ct+1

)1−σ

(6)

Among advantages associated with the use of linearly homogeneous representation of preferences
(4a), it is simplifies reporting of welfare changes. Hicksian-equivalent variations is trivially
measured with CES: a 1% change in Û(.) corresponds to a 1% equivalent variation in income.
Thus we will relate the Ramsey intertemporal preferences to CES utility functions which are
commonly employed in static general equilibrium models.

3 Ramsey’s Equilibrium Problem

Equilibrium in the Arrow-Debreu model is characterized by three classes of equations mentioned
in Section 2. To be able to formulate those equations in terms of MCP modelling we have to define
also complementarity pairs. To illustrate it, we implement a specific policy experiment with
Ramsey model, letting τt represent the capital income tax applied in year t.

Unlike the formulation (4) and many macroeconomics textbooks that express aggregate output Yt
as a function of capital stock alone: Yt = f(Kt), we will use two production factors. In our

representation of the Ramsey model, it is convenient to work with a constant-returns production
function in which we have inputs of both labor and capital:

Yt = F (Kt, L̄t) (7)

When labor is in fixed supply (L̄t), the production function exhibits diminishing returns to
capital. There is therefore no loss of generality by formulating the model on the basis of a
constant returns to scale technology.

(i) Market clearance conditions and associated market prices3

• Output market: Yt = Ct(pt,M) + It ⊥ pt ≥ 0

• Labor market: L̄t = aL(r
K
t (1 + τt), p

L
t ) Yt ⊥ pLt ≥ 0

• Market for capital services: Kt = aK(rKt (1 + τt), p
L
t ) Yt ⊥ rKt (1 + τt) ≥ 0

• Capital stock: Kt+1 = (1− δ)Kt + It ⊥ pKt ≥ 0

where pt, p
L
t , r

K
t , and pKt represents market price, wage rate, capital rental rate, and capital

purchase price respectively. The compensated demand functions for labor (aL) and capital
(aK) are the partial derivatives of the unit cost function.

(ii) Zero profit conditions and associated activities

• Production process: pt = c(pLt , r
K
t (1 + τt)) ⊥ Yt ≥ 0

• Investment: pt ≥ pKt+1 ⊥ It ≥ 0

• Capital stock: pKt = rKt + (1− δ)pKt+1 ⊥ Kt ≥ 0

where c(.) is a price index for value-added. The only activity level which could possibly fall
to zero would be investment, and that would only happen in a policy scenario which
resulted in a substantial reduction in the return to capital.

(iii) Income balance

• M = pK0 K̄0 +
∑∞

t=0

(

pLt L̄t + τtr
K
t Kt

)

− pKT+1KT+1

The choice of T depends on the nature of the policy experiment and the speed with which the
model converges to a new steady-state growth path. A benchmark steady-state can be reached
through a calibration technique. The idea is that if a model is parameterized on the basis of
economic transactions in a reference equilibrium with policy parameters τ̄ , then the model should
be able to reproduce that equilibrium when given policy parameter τ = τ̄ . Such consistency check
is required for any CGE modeling.

3The demand functions employed in this model assure that all prices will be nonzero in equilibrium. There is no
formal need, therefore, to associated prices with market clearance conditions, as would be required in a conventional
complementarity problem. We provide an associated here in order to help understand how the model might be
extended with demand functions which would admit zero prices.

4 Calibration technique for infinite horizon

Benchmark replication is a common strategy for evaluating logical consistency of both static and
dynamic general equilibrium models. An immediately evident challenge which Ramsey’s model
presents for economists commonly working with static models is the infinite time horizon. We are
only able to process models with a finite number of variables. When moving from static to
dynamic models, we move from computing the exact equilibrium to approximating the
infinite-horizon transition path using tools for solving finite dimensional optimization and
complementarity problems.

6 6 6

� -

� -

t

tt

tf(t) tl(t)

tp(tt)

2009 . . . 2100 2101

tt Time horizon (with the first year of the post-terminal period)
t(tt) Time period over the model horizon
tf(tt) First period of the model (2009)
tl(tt) Last endogenous period (2100)
tp(tt) Post-terminal period (the element of set tt which is not an element of t, 2101)

Figure 1: Sets Defining Time Structure of the Ramsey model

How an infinite horizon model can be effectively represented by a T -period approximation has
been explained by Lau et al. (2002). The solution process requires to decompose the consumer’s
problem: a one run through time period T and another for the post-terminal period (T + t). The
Figure 1 shows a relationship between time periods over the model horizon.

If a growth model is calibrated to a steady-state growth path, this can serve the same role as the
benchmark equilibrium dataset in a static model. One problem with this strategy is that in many
cases, the base year data may be far removed from a steady-state growth path. In this situation,
as a case, it may be possible to adjust the database (only for the purpose of debugging the model
formulation) such that some versions of the model are consistence with the steady-state growth.

It is customary, however, first to solve a business-as-usual growth path as a reference case prior to
solving other cases, so policy analysis consists of comparing one adjustment path with another.

Here are the steps involved in sorting out the stead-state conditions with related investments and
capital earnings in a static data set which is consistent with a steady-state growth path.

(i) Imputed Capital Cost:

The simplest approach is to set up the model along a steady-state growth path in which
interest rate (r̄) and growth rate (ḡ) are given and assumed constant. The zero-profit
condition for investment reveals the price level for capital (pKt ):

pKt+1 =
pKt
1 + r̄

= Pt (8)

The base year price of capital, assuming the Harberger convention (Harberger, 1962), is
then: p̄K0 = 1 + r̄.

(ii) Marginal Product of Capital:

The zero profit condition for capital determines the capital rental price (rKt ):

pKt = rKt + (1− δ)pKt+1 (8a)

Note that in a static modelling we do not have pKt , but only rKt . Substituting the values of
pKt and pKt+1 reveals that the base year rental price of capital is sufficient to cover interest
plus depreciation: r̄K0 = r̄ + δ.

(iii) Macro Reconciliation:

Finally, consider the market clearance condition for capital in the first period:

K1 = K̄0(1− δ) + Ī0 = (1 + g)K̄0 (8b)

This implies that base year investment can be calculated on the basis of growth and
depreciation of the base year capital stock: Ī0 = K̄0(g + δ). We then can use r̄K0 to
determine K̄0 on the basis of the value of capital earnings in the base year (V̄ K

0 ):

Ī0 = V̄ K
0

g + δ

r + δ
(8c)

The problem that arises in applied models is that Ī0 and V̄ K
0 = K̄0r̄

K
0 will not satisfy this

relation for arbitrary values of ḡ, r̄ and δ. Part of the art of economic equilibrium analysis
involves reconciliation of theory and data. It is perhaps not surprising that when we add
more theory (e.g., intertemporal decisions) we introduce more constraints on the underlying
database.

The terminal approximation in a model with a single consumer is quite simple, because we do not
need to define a value of terminal assets. A precise approximation only requires that the terminal

capital stock closely follows the steady-state value. This is consistent with the convergence in
growth rates of all macroeconomic variables, the most sensitive of which is investment.
Post-terminal capital stock, KT+1, can be chosen such that the terminal growth rate of capital
matches aggregate economic growth.

In models with multiple consumers living beyond period T , it would be necessary to account for
which of these agents owns the assets. Some agents may have negative asset positions at the end
of the model (particularly in overlapping generations models where young households accumulate
debt which is repaid in middle age).

5 Optimization versus Complementarity

The NLP models have lower precision compared to alternative models formulated with MCP. For
example, Lau et al. (2002) shows that Ramsey model formulated with NLP substantially
underestimate the level of debt. Complementarity determines what activities run, who produces
what, and who consumes what. The cost of this precision is a requirement explicitly represent
both primal and dual optimality conditions. Optimization approach is more compact and it
assumes that we control the complete system. In order to use the advantages of both approaches,
we apply a new tool EMP discussed in section 1.

5.1 Extended Mathematical Programming

Extended Mathematical Programming allows to define a model in a compact way (NLP) and use a
precision of complementarity approach (MCP). Adjusting optimization problem (4) to a policy
experiment with capital income tax described in section 3, the Ramsey optimization problem (3)
looks as following:

max
Ct,It,Kt|KT+1,r

K
t

(

1

1 + ρ

)t C1−θ
t − 1

1− θ
−
∑

t

τtr
K
t Kt (9)

s.t. ψl1−α
t kαt = Yt = It + Ct, Kt+1 = (1− δ)Kt + It, kt = Kt ⊥ rKt , lt = L̄t

where the auxiliary variable KT+1 and the associated auxiliary equation IT /IT−1 = 1 + g
implements the steady-state condition for terminal investment growth.4 The model is initially
calibrated to a steady-state growth path with no capital income tax (τt = 0). Later we evaluate
the economic consequences of such taxation assuming lump-sum revenue recycling to the
representative agent.

The EMP formulation (optimization problem) can be defined in primal or dual form. In economic
analysis, preferences may be described by a primal form utility function, U(C), or, equivalently,

4This termination technique is hence characterized as state variable targeting. A state variable’s terminal value is
chosen such that the growth rate of the associated control variable is consistent with steady-state growth.

using a dual form indirect utility function, V (p,M) =M/e(p). When preferences are homothetic,
the unit expenditure function (5a) conveys all of the information concerning the underlying
preferences. Closed form expressions for U(.) and V (.) are not always available. The most flexible
functional forms have a dual form but no explicit primal form. The constant elasticity functions
used in the present analysis admit both primal and dual forms. Thus the Ramsey maximization
problem (9) can be represented using the expenditure function in place of the utility function:

min
p,pL,pK ,rK |I,K

∑

t

pLt Lt + pK0 K0 − pKT+1KT+1 + τtr
K
t Kt − M̄e(p) (10)

s.t. (pLt )
1−α(rKt (1 + τt))

α =pt, pKt = pKt+1(1− δ) + rKt ⊥ Kt, pt ≥ pKt+1 ⊥ It

The constraints for the dual optimization problem are the zero profit conditions for the
underlying equilibrium problem. The income balance is included into the objective function (10).
The market clearance conditions for the equilibrium are incorporated in the first-order optimality
conditions. According to Sheppard’s lemma, the demand in year t equals the utility level times
the gradient of the expenditure function: Ct(p,M) = V (p,M)∂e(p)/∂pt. Substituting into the
first order condition, we see that the optimality conditions for the dual correspond to the market
clearance conditions which appears explicitly in the primal.

5.2 EMP and MCP with GAMS

Primal and dual forms with EMP require to distinguish between primal decision variables and
dual variables. Complementary pairs should be defined only for dual variables in the primal EMP
formulation and for primal variables in the dual EMP formulation. Also auxiliary constraints
requires directly to determine a complementary variable. All other variables do not need
complementary pairs with EMP, but primal (dual) multipliers should be assigned to equations in
dual (primal) forms. On the other hand, the MCP formulation (equilibrium problem) requires
complementary pairs for all variables.

In the case of Ramsey equilibrium problem, there are six primal variables Yt, It,Kt, Ct, LDt,KDt,
four dual variables pt, r

K
t , p

K
t , p

L
t , income variable, and auxiliary variable. The GAMS code is

available in Appendix A1.5 The same model formulated as a primal EMP model (Apeendix A2)
has only one dual varible (capital rental price rKt ), seven primal variables, and decision variable.
The additional to MCP primal variables is utility level.

The EMP framework is available with GAMS software. This framework requires to declare
complementary pairs in a separate file. This information file redefines NLP formulation into
MCP. This means that the information file for Ramsey model should contain a declaration for one
implicit auxiliary variable-equation pair and one explicit dual variable-equation pair. The explicit

5We use a GAMS tool $macro that allows to formulate some equations with no declaration of complementarity
pairs. These equations (cost function, compensated factors demand, consumption function) just help to formulate
necessary equilibrium conditions described in Section 3.

dual variable, rKt , is introduced in order to provide an explicit measurement of capital earnings.
The associated equation for this dual variable is a market clearing condition for capital service.
The implicit auxiliary variable is a post-terminal capital stock KT+1. The associated auxiliary
equation is explicitly defined.

Next, an option file should be created in order to save results from reformulation of NLP
framework into MCP. Finally, we verify calibration using zero iterations. If the baseline growth
path is replicated, then we able to simulate scenarios. Our experiment is based on the scenario,
where capital income tax τt = 25% is applied in year t = 6. For this purpose the information file
should be redefined.

Thus in a primal EMP formulation only market clearing conditions are applied explicitly, while
MCP format requires to define all necessary conditions. Instead of zero profit condition and
income balance we rather have to define production and utility functions with primal EMP.6 In
order to get a positive Lagrange multiplier, the objective function includes a utility function with
a negative sign. This means that we have to minimize the objective function in the primal EMP
model in order to get the same results as in the MCP model. Since our simulation experiment
includes a lump-sum recycling of capital income tax revenue, the objective function includes also
that revenue.

Alternatively, the dual EMP formulation requires to declare only zero-profit condition (Appendix
A3). Market clearing condition is replaced by expenditure function and income balance is
replaced by objective function. Here we have five dual variables, two primal variables, and
decision variable. The information file includes complementary pairs only for auxiliary equation
and for primal variables. Since auxiliary constraint refers to investment in both periods IT and
IT−1, then we have to declare a separate complementary pairs for investments in both periods.

5.3 Application of extended mathematical programming

5.3.1 Ramsey model with tax on capital earnings

We have assigned a permanent tax on capital earnings (RKt ∗Kt) of 25% beginning in 2013 using
a model described in Sections 3 and 5.1. Cobb-Douglass production structure allows to keep the
perfect foresight model simple. The equilibrium is computed assuming that economic agents
anticipate the application of the tax, resulting in a sharp response in investment and other
economic variables to the new economic environment. Exactly the same results were obtained by
MCP and both EMP versions.

Selected results are presented in Figure 2. When economic agents’ behavior depend on
forward-looking expectations, then a credible announcement of a future change in policy has an
effect already before the new policy is implemented. Immediately after 2013, investment will be

6No $macro tool can be used to define these functions.

Figure 2: Simulation of tax on capital in year 5 (Ramsey MCP Model)

decreasing because saving is less than required to sustain a constant capital stock and
consumption will be slightly increasing. At time 2013 (year 5) the economy reaches a new saddle
path, the high taxation on capital income begins, and the rate of return becomes considerably
lower. Future consumption becomes relatively expensive and elasticity of intertemporal
substitution cause a shift in consumption from the future to the present. The economy gradually
approaches to the new steady state.

The Walras law implies that capital flows may be positive in some periods or negative in others,
but the sum of their present values is zero over the planning horizon (Manne and Rutherford,
1994). Over time the tax leads to a smooth reduction in the steady-state capital stock, while
capital purchase price returns to steady state after 44 years. Investment is typically the most
sensitive item, it decrease by over 30% in 2013 and then decrease smoothly, while capital rental
price returns to a steady state after 69 years. Tax on capital earnings leads to smooth decrease of
labor price, because labor supply has exogenous growth path, while capital is adjusted
endogenously.

5.3.2 Negishi model with sales tax

Negishi (1960) looked for the set of Pareto efficient allocations (as solutions of a central planner’s
problem) to show (using Second Welfare Theorem) that there exists an allocation which
constitutes a general equilibrium. Solving a planner’s problem is easier than directly obtaining
computations of competitive equilibria. In doing so we have to be careful in applying optimal
control theory because there are exists difficulties with respect to differentiability and boundness
of equilibrium allocations (van Geldrop and Withagen, 1994). However, when the number of
commodities is large relative to the number of consumers, then the excess demand approach
(Scarf, 1973) is inefficient compared to the Negishi approach.7

7Ginsburgh and Keyzer (2002, p.113) provides the following explanation for such inefficiency. Every agent in the

If the economy is Walrasian, then all equilibrium allocations are Pareto efficient. Otherwise, the
parametrization used to characterize equilibrium allocations will fail to capture inefficient
equilibria. For examples see Tirelli (2010). A competitive equilibrium can be also represented
through a welfare optimum with nonzero welfare weights, which satisfy budget constraints for all
households. Negishi’s approach assumes arbitrary weights for consumers. Ginsburgh and van der
Heyden (1988) shows how to represent market imperfections (such as taxes or maximum prices)
in the Negishi format. Such allocation can be found by maximizing a weighted sum of the welfare
functions and government operations subject to the technological and system constraints:

max
C,X|θ,p

∑

h

θh lnUh(Cg,h) +
∑

g

tgpgYg(Xf,g)

s.t.
∑

h

Cg,h ≤ Yg(Xf,g),
∑

g

Xf,g ≤
∑

h

ēh,f (11)

where welfare weights θh =
∑

f eh,f/
∑

h

∑

f eh,f are interpreted as Negishi coefficients (weights).
They depend on fixed factor endowment eh,f and they hold the budget constraint for every
household h. Here sales tax tg is proportional to output value, while output price pg is a vector of
Lagrange multipliers associated with market clearing condition for goods g.

There are three ways to solve this model with taxes. If we use NLP method, we have to apply a
sequential joint maximization approach (Rutherford, 1999) that (i) solves the NLP with no tax
using the fixed prices and weights for primal NLP or fixed income and weights for dual NLP.8 (ii)
updates the prices based on the new marginal of market clearing conditions for both factors and
goods from the NLP solution with tax; update the weights based on equilibrium income (iii)
computes the difference between old and new weights until the difference is small enough. As the
weights converge, the agents will move toward balanced budgets, where their incomes equal to
their expenditures. This means that no income balance can be formulated with a primal NLP
model because no endogenous price of factors in a primal NLP. Instead, we use a loop with two
additional variables: factor price wf and equilibrium income
Mh =

∑

f wfeh,f + 1/cardh
∑

g tgpgYg.

Let’s consider a 2x2x2 model in which a set of consumers (H1,H2) are each endowed with a fixed
quantity of factors (F1, F2). The first household posses only F1 and the second household - F2.
The solution consists of a consumption vector for each agent and a set of prices for each good
(G1, G2) such that each agent maximizes her utility at this consumption level. Utility is given by
a CES function and production technology is described by Cobb-Douglass function. In order to
initialize the weights, they were chosen randomly (see Appendix 4). The scenario with a sales tax
on first good t(G1) = 0.5 implies that income of the first household goes up because the price of
F1 decrease less than F2. Production of G1 requires relatively less F1, thus a sales tax implies a

excess demand format (fixed point approach) solves his own welfare function and the feedback is an adjustment on
prices only. While in the Negishi approach, there is a single welfare function and the feedback is an adjustment on
prices and weights.

8Weights and prices should be fixed with primal NLP because these variables define market clearing conditions
for factors and goods respectively. Weights and supply should be fixed with dual NLP because these variables define
zero profit conditions for factors and goods producers respectively.

Figure 3: Simulation of sales tax on good G1 (Negishi NLP Model)

relatively less decrease in price of F1. As a result, the Negishi coefficient for H1 is higher than
H2(Figure 3).

Welfare adjustment process enables the computation of an equilibrium, adjusting the weights
assigned to each individual in the social welfare function. We can confirm this result using a
second method to solve this model - EMP. It is allow to find the Negishi coefficients directly. The
objective function and primal constraints are the same as NLP model, but side constraints for
welfare coefficients and income should be also included directly into the main body of the model.
Then for dual variables and side constraints we have to assign complementary pairs: wf for
factors market clearing condition and pg for goods market clearing condition, income share for θh,
and income definition for Mh.

Both codes for primal NLP and EMP Negishi model are available in Appendix 4. The dual
representation is alike. NLP approach requires few iterations (Figure 3) to get the same solution
as EMP. Solution speed is five times greater with EMP than NLP.

6 Conclusion

Equilibrium problem described by MCP can be interpreted as first-order conditions for
optimization problem described by NLP. The disadvantage of the MCP approach is a model
specification, because complementarity requires explicit representation of both optimality
conditions (primal and dual). It might be difficult to design such structure for a model. The
example is the DICE model. In this case the EMP framework helps. This paper outlines that
approach within GAMS modeling system for an analysis of interactions among forward-looking
players in dynamic environment (Ramsey model).

Optimization approach assumes that we control the complete system and the model
representation is compact. Three classes of the equilibrium conditions (market clearing, zero

profit, and income balance) required in the complementarity framework are reduced just to one
class in the optimization framework (market clearing for primal approach or zero profit for dual
approach). Instead of zero-profit conditions and income balance, we have to formulate production
function, utility function, and objective function in the primal optimization approach. Instead of
market-clearing conditions and income balance, we have to formulate cost function, expenditure
function, and objective function in the dual optimization approach.

Mathematical programs with optimization problems in their constraints have a long history in
operations research. Yet Ferris and Sinapiromsaran (2000) found that forming NLP model and
solving this as a complementarity problem is a viable approach. With the extended mathematical
programming it works well, according to our tests for three economic models. EMP allows for
computational advances, such as automatic differentiation technique, to eliminate errors that
occur without automation and to conduct models with formulations otherwise too
time-consuming to consider. Specifically:
1) NLP formulation is compact compared to MCP. Thus EMP allows to use advantage of NLP to
formulate a model.
2) NLP solvers uses quite high memory and superbasic variables. MCP does not require
superbasics. Thus EMP allows to solve better large models.
3) NLP framework requires that the model horizon is sufficiently long to converge to the
steady-state. MCP framework better approximate the infinite horizon and it is virtually
insensitive to the model horizon (Boehringer et al., 2007). Thus EMP allows to improve model
robustness.
4) we have to be careful in applying optimal control theory because there are exists difficulties
with respect to differentiability and boundness of equilibrium allocations. Thus it better to apply
the First Welfare Theorem (equilibrium approach) rather than the Second one (optimization).
5) MCP format accommodates a wider range of economic complexities, than NLP (more goods,
more regions, tax revenue recycling, etc.)
6) MCP relaxes the integrability constraints (the shadow prices coincide with market prices)
imposed by the NLP framework. Thus EMP allows to accommodate second-best settings that
reflect initial inefficiencies.

Every NLP model can be represented with MCP framework but not vice versa. Complementarity
slackness conditions determine which prices and activity levels are at zero level, and which are
positive. We use a bridge (EMP framework) that allows to formulate a model as an optimization
problem, but solve it as a complementarity problem. In the Ramsey experiment we have verified
whether MCP and EMP produce the same results using a simulation of capital income tax. In the
Negishi model we have analysed whether NLP and EMP produce the same results using a
simulation of sales tax.

Appendix A: GAMS Code

A.1: The MCP approach for Ramsey Model

$TITLE Ramsey Model -- MCP formulation calibrated to base year data

$include dataCES

Nonnegative Variables

* Activity levels:

Y(t) Output

I Investment

K Capital stock

* Prices:

P Output price

RK Rent of capital

PK(tt) Capital price

PL(t) Wage rate

* Income level:

RA Representative agent

* Auxiliary variable (targeting of steady-state):

TK Post-terminal capital stock

* Cobb-Douglas Price index for value-added:

$macro PKL(t) (PL(t)**(1-kvs) * {RK(t)*(1+taxk(t))}**kvs)

* Cobb-Douglas Compensated labor demand:

$macro LD(t) (l0 * PKL(t)/PL(t))

* Cobb-Douglas Compensated capital demand:

$macro KD(t) (kd0 * PKL(t)/{RK(t)*(1+taxk(t))})

* CEIS Consumption demand:

alias (t,t_);

$macro expend (sum(t_, thetac(t_) * (P(t_)/pref(t_))**(1-sigma))**(1/(1-sigma)))

$macro CD(t) (qref(t)*c0 * (RA/(expend*ra0)) * (expend*pref(t)/P(t))**sigma)

Equations

* Market clearing condition

markety(t) goods market,

marketrk capital market

marketl(t) labor market

marketk investment market

* Zero profit condition

profity(t) production,

profiti(tt) investments

profitk capital

* Income balance

income representative agent budget

* Auxiliary variables:

tkdef post-terminal capital stock;

markety(t).. Y(t)*y0 =e= I(t)*i0 + CD(t);

profity(t).. PKL(t)*(l0 + kd0) =e= P(t)*y0;

marketl(t).. qref(t)*l0 =e= Y(t)*LD(t);

marketk(tt).. K(tt-1)*k0*(1-delta) + I(tt-1)*i0 + k0$tf(tt) =e= K(tt)*k0$t(tt) + TK$tp(tt);

marketrk(t).. K(t)*kd0 =e= Y(t)*KD(t);

profitk(tt)$t(tt).. PK(tt)*k0 =e= PK(tt+1)*k0*(1-delta) + RK(tt)*kd0;

profiti(tt)$t(tt).. P(tt) *i0 =e= PK(tt+1)*i0;

tkdef.. sum(tl(t), I(t)/I(t-1) - Y(t)/Y(t-1)) =e= 0;

income.. RA =e= sum(t, PL(t)*qref(t)*l0)

+ sum(tf, PK(tf)*k0)

- sum(tp, PK(tp)*TK)

+ sum(t, taxk(t)*RK(t)*KD(t)*Y(t)) ;

model ramseym /markety.P, profity.Y, marketl.PL, marketk.PK, marketrk.RK, profitk.K, profiti.I,

tkdef.TK, income.RA/;

* Assign steady-state equilibrium values for quantities

* and prices:

Y.L(t) = qref(t);

I.L(t) = qref(t);

K.L(t) = qref(t);

P.L(t) = pref(t);

RK.L(t) = pref(t);

PL.L(t) = pref(t);

* The steady-state price of capital is the output price

* times one plus the interest rate:

PK.L(tt) = (1+r) * pref(tt);

TK.L = k0 * (1+g)**card(t);

RA.FX = ra0;

ramseym.iterlim = 0;

solve ramseym using mcp;

ABORT$(ramseym.objval > 0.001) "Model does not calibrate.";

* Apply a tax on capital inputs of 25% beginning in year 6:

taxk(t)$(t.val > 2012) = 0.25;

ramseym.iterlim = 100000;

solve ramseym using mcp;

* Extract consumption as a variable for use in the report:

variable C(t) Consumption;

$ondotl

C.L(t) = CD(t);

$include report

A.2: The Primal EMP approach for Ramsey Model

$TITLE Ramsey Model -- EMP primal formulation calibrated to base year data

$include dataCES

parameter psi(t) Production function scale factor

rho Primal form elasticity;

psi(t) = y0 /(l0**(1-kvs) * kd0**kvs);

rho = 1 - 1/sigma;

Variable OBJ Objective function;

Nonnegative Variables

* Primal decision variables:

Y(t) Output

I Investment

K Capital stock

C Consumption

KD Capital demand

LD Labor demand

W utility

* Dual (auxiliary) variables used in the EMP framework:

RK(tt) Rental price;

Equations

* Market clearing condition:

markety(t) goods market,

marketrk capital market

marketl(t) labor market

marketk investment market

* Functions:

production(t) production function

objdef objective function

welfare utility function

* Auxiliary constraint:

tkdef post-terminal capital stock;

objdef.. OBJ =e= -W + sum(t, taxk(t)*RK(t)*KD(t));

welfare.. ra0 * sum(t, thetac(t)*(C(t)/(qref(t)*c0))**rho)**(1/rho) =e= W;

production(t).. psi(t) * LD(t)**(1-kvs) * KD(t)**kvs =e= Y(t);

markety(t).. Y(t) =e= I(t) + C(t);

marketl(t).. qref(t)*l0 =e= LD(t);

marketrk(t).. K(t)*rk0 =e= KD(t);

marketk(tt).. K(tt-1)*(1-delta) + I(tt-1)+ k0$tf(tt) =e= K(tt);

tkdef(tl(t)).. I(t)/I(t-1) =e= Y(t)/Y(t-1);

* Declare the model with all of the equations defined here:

model ramsey /all/;

* Assign steady-state values:

C.L(t) = qref(t)*c0;

Y.L(t) = qref(t)*y0;

I.L(t) = qref(t)*i0;

K.L(tt)= qref(tt)*k0;

KD.L(t)= qref(t)*kd0;

LD.L(t)= qref(t)*l0;

RK.L(t)= pref(t);

W.L = ra0 ;

OBJ.L = -ra0;

* Assign the dual multipliers consistent with this growth path:

production.m(t) = pref(t);

markety.m(t) = pref(t);

marketrk.m(t) = pref(t);

marketl.m(t) = pref(t);

marketk.m(tt) = pref(tt)*(1+r);

welfare.m = 1;

* Write an information file which indicates which auxiliary

* variables and constraints:

file empinfo / ’%emp.info%’ /;

put empinfo;

loop(t$taxk(t), put / ’dualvar ’ RK.tn(t) ’ ’ marketrk.tn(t));

loop((tl,tp), put / ’dualequ ’ tkdef.tn(tl) ’ ’ K.tn(tp));

putclose;

* Generate an EMP option file requesting that the scalar

* MCP file and a dictionary both are saved:

$onecho >emp.opt

filename=emp-p-scalar.gms

Dict=emp-p-scalar.txt

$offecho

ramsey.optfile = 1;

* First solve has no iterations to replicate the baseline growth path:

ramsey.iterlim = 0;

SOLVE ramsey USING emp minimizing OBJ;

* Apply a tax on capital inputs of 25% beginning in 2013:

taxk(t)$(t.val > 2012) = 0.25;

* Regenerate the EMP control file which changes when taxk is modified:

put empinfo;

loop(t$taxk(t), put / ’dualvar ’ RK.tn(t) ’ ’ marketrk.tn(t));

loop((tl,tp), put / ’dualequ ’ tkdef.tn(tl) ’ ’ K.tn(tp));

putclose;

* Solve the counterfactual:

21



ramsey.iterlim = 10000;

SOLVE ramsey USING emp minimizing OBJ;

variable P(t) Output price,

PK(t) Capital price,

PL(t) Wage rate;

P.L(t) = markety.m(t);

PK.L(t)= marketk.m(t);

PL.L(t)= marketl.m(t);

RK.l(t)= marketrk.m(t);

I.L(t) = I.L(t)/i0;

K.l(t) = K.L(t)/k0;

$include report

A.3: The Dual EMP approach for Ramsey Model

$TITLE Ramsey Model -- EMP dual formulation calibrated to base year data

$include dataCES

Variables

OBJ Objective function;

Nonnegative Variables

* Dual variables

RK(tt) Rental rate

PL Labor price

P Output price

PK(tt) Capital price,

EX Expenditures

* Primal variables

I(tt) Investment

K(tt) Capital stock;

Equations

* Zero profit condition

profity(t) production,

profiti(tt) investments

profitk capital

* Functions

expend expenditures function

objdef objective function

* Auxiliary constraint

tkdef post-terminal capital stock;

objdef.. OBJ=e= sum(t, PL(t)*qref(t)*l0)

+ sum(tf, PK(tf)*k0)

- sum(tp, PK(tp)*K(tp))

+ sum(t, taxk(t)*RK(t)*rk0*K(t))

- EX;

expend.. ra0 * sum(t, thetac(t) * (P(t)/pref(t))**(1-sigma))**(1/(1-sigma)) =e= EX;

profity(t).. PL(t)**(1-kvs) * (RK(t)*(1+taxk(t)))**kvs =e= P(t);

profitk(t(tt)).. PK(tt) =e= PK(tt+1)*(1-delta) + RK(tt)*rk0;

profiti(t(tt)).. P(tt) =e= PK(tt+1);

tkdef(tl(t)).. I(t)/I(t-1) =e= 1+g;

model ramseyd /profity, profiti, profitk, objdef, expend, tkdef /;

OBJ.L = 0;

I.L(t) = i0*qref(t);

K.L(tt) = k0*qref(tt);

RK.L(t) = pref(t);

PK.L(tt)= pref(tt)*(1+r);

P.L(t) = pref(t);

PL.L(t) = pref(t);

EX.FX = ra0;

* Assign the primal multipliers consistent with this growth path:

expend.m = 1;

profity.m(t) = y0*qref(t);

profitk.m(tt) = k0*qref(tt);

profiti.m(t) = i0*qref(t);

* Translate NLP into MCP

file dempinfo / ’%emp.info%’ /;

put dempinfo;

loop((tl,tp), put / ’dualequ ’ tkdef.tn(tl) ’ ’ K.tn(tp));

loop(t$taxk(t), put / ’dualvar ’ K.tn(t) ’ ’ profitk.tn(t));

loop(tl(t), put / ’dualvar ’ I.tn(t) ’ ’ profiti.tn(t)

/ ’dualvar ’ I.tn(t-1) ’ ’ profiti.tn(t-1));

putclose;

$onecho >emp.opt

filename=emp-d-scalar.gms

Dict=emp-d-scalar.txt

$offecho

ramseyd.optfile = 1;

* Replicate the benchmark equilibrium:

ramseyd.iterlim = 0;

SOLVE ramseyd USING emp minimizing OBJ;

* Apply a tax on capital inputs of 25% beginning in year 6:

taxk(t)$(t.val > 2012) = 0.25;

* Regenerate the information file which changes when taxk increases from zero:

put dempinfo;

loop((tl,tp), put / ’dualequ ’ tkdef.tn(tl) ’ ’ K.tn(tp));

loop(t$taxk(t), put / ’dualvar ’ K.tn(t) ’ ’ profitk.tn(t));

loop(tl(t), put / ’dualvar ’ I.tn(t) ’ ’ profiti.tn(t)

/ ’dualvar ’ I.tn(t-1) ’ ’ profiti.tn(t-1));

putclose;

* Solve the model:

ramseyd.iterlim = 1000;

SOLVE ramseyd USING emp minimizing OBJ;

variable C(t) Consumption;

* Extract solution values for capital stock, investment and output:

24



I.L(t) = profiti.m(t)/i0;

K.l(t) = profitk.m(t)/k0;

C.L(t) = profity.m(t) - profiti.m(t);

$include report

A.4: Negishi Model with sales tax

The Primal NLP approach

$title A Simple Negishi model using primal NLP optimization

$include data_negishi.gms

Variables U utility,

OBJ objective;

nonnegative variables

C(g,h) Household consumption,

X(f,g) Factor inputs,

Y(g) Sectoral supply,

P(g) Market price estimate,

THETA(h) Negishi coefficients (welfare weights);

Equation objdef objective function,

udef utility function,

market good market clearing condition,

supply production function,

fmarket factor market clearing condition;

objdef.. OBJ =e= -sum(h,u0(h))*sum{h, THETA(h) * log(U(h))} + sum(g, t(g)*P(g)*Y(g));

udef(h).. sum(g, alpha(g,h) * (C(g,h)/c0(g,h))**rho)**(1/rho) =E= U(h);

market(g).. Y(g) =e= sum(h, C(g,h));

supply(g).. phi(g) * prod(f, X(f,g)**beta(f,g)) =e= Y(g);

fmarket(f).. sum(h, e(h,f)) =e= sum(g, X(f,g)) ;

* No income balance should be declared because it will include dual variable W(f),

* but for Primal approach we do not define dual variables (except P(g) that is fixed)

model negishi_nlp /objdef, udef, supply, fmarket, market/;

$ondotl

C.lO(g,h) =0.001;

X.lO(f,g) =0.001;

U.lO(h) =0.001;

Y.L(g) =y0(g);

C.l(g,h) =c0(g,h);

X.l(f,g) =x0(f,g);

U.l(h) = 1;

* P(g) should be fixed because it defines ’market’ equation

P.FX(g) = 1;

* THETA should be fixed because it defines ’fmarket’ equation

THETA.FX(h)=sum(f, e(h,f))/sum(hh,sum(f, e(hh,f)));

t(g) = 0;

* Load the solver into memory to be able to iterate more

* rapidly at the expense of occupying a bit of memory:

negishi_nlp.solvelink = 2;

*execute_loadpoint ’MGEMDL_p.gdx’;

*negishi_nlp.iterlim = 0;

Solve negishi_nlp minimizing OBJ using nlp;

t("brd") = 0.5;

* To make a simulation for NLP we need to create a loop, because THETA.FX and P.FX

nonnegative variables

W(f) factor price

INCOME equilibrium income ;

set iter Negishi iterations /iter0*iter20/;

parameter itrlog Iteration log;

* Randome starting point:

INCOME.FX(h)= uniform(0,100); ;

loop(iter,

P.FX(g) = market.m(g);

THETA.FX(h) = INCOME.L(h)/sum(h.local,INCOME.L(h));

itrlog(iter,h) = THETA(h) * 100;

itrlog(iter,g) = Y(g);

solve negishi_nlp using nlp minimizing OBJ;

W.FX(f) = fmarket.m(f);

INCOME.FX(h)= sum(f, W(f)*e(h,f)) + (1/card(h)) * sum(g, P(g)*Y(g)*t(g)););

display itrlog, W.L, INCOME.L;

$setglobal domain iter

set iterlbl(iter) /iter0 0, iter5 5, iter10 10, iter15 15, iter20 20/;

$setglobal labels iterlbl

$libinclude plot itrlog

The Primal EMP approach

$title A Simple Negishi model using primal EMP optimization

$include data_negishi.gms

Variables U utility,

OBJ objective;

nonnegative variables

C(g,h) Household consumption,

X(f,g) Factor inputs,

Y(g) Sectoral supply,

* Dual variables introduced through side constraints:

P(g) Market price estimate,

W(f) factor price

INCOME(h) Equilibrium income

THETA(h) Negishi coefficients;

Equation objdef objective function,

udef utility function,

market good market clearing condition,

supply production function,

fmarket factor market clearing condition,

incomedef Defines income,

thetadef Defines income share;

objdef.. OBJ =e= -sum(h,u0(h))*sum{h, THETA(h) * log(U(h))} + sum(g, t(g)*P(g)*Y(g));

* Primal constraints which would appear in the NLP model:

udef(h).. sum(g, alpha(g,h) * (C(g,h)/c0(g,h))**rho)**(1/rho) =E= U(h);

market(g).. Y(g) =e= sum(h, C(g,h));

supply(g).. phi(g) * prod(f, X(f,g)**beta(f,g)) =e= Y(g);

fmarket(f).. sum(h, e(h,f)) =e= sum(g, X(f,g));

* Side constraints for the EMP model (we do not need it for NLP):

incomedef(h).. INCOME(h) =e= sum(f, W(f)*e(h,f)) + (1/card(h)) * sum(g, P(g)*Y(g)*t(g));

thetadef(h).. THETA(h)*sum(hh,INCOME(hh)) =e= INCOME(h);

model negishi_emp /objdef, udef, supply, fmarket, market, incomedef, thetadef/;

file olga / ’%emp.info%’ /;

C.lO(g,h) =0.001;

X.lO(f,g) =0.001;

U.lO(h) =0.001;

Y.L(g) = y0(g);

C.l(g,h) =c0(g,h);

X.l(f,g) =x0(f,g);

P.L(g) = 1;

W.L(f) = 1;

INCOME.L(h) = sum(f,e(h,f));

* Distribute Negishi coefficients uniformly (starting point)

THETA.L(h) = 1/card(h);

U.l(h) = 1;

* Replicate the benchmark equilibrium -- all prices should be unity:

t(g) = 0;

put olga ;

loop{f, put / ’dualvar ’ W.tn(f) ’ ’ fmarket.tn(f)};

loop{g$t(g), put / ’dualvar ’ P.tn(g) ’ ’ market.tn(g)};

loop{h, put / ’dualequ ’ thetadef.tn(h) ’ ’ THETA.tn(h)};

loop{h, put / ’dualequ ’ incomedef.tn(h)’ ’ INCOME.tn(h)};

putclose;

Solve negishi_emp minimizing OBJ using emp;

* Solve a counter-factual in which we apply a 50% excise tax:

t("brd") = 0.5;

put olga ;

loop{f, put / ’dualvar ’ W.tn(f) ’ ’ fmarket.tn(f)};

loop{g$t(g), put / ’dualvar ’ P.tn(g) ’ ’ market.tn(g)};

loop{h, put / ’dualequ ’ thetadef.tn(h) ’ ’ THETA.tn(h)};

loop{h, put / ’dualequ ’ incomedef.tn(h)’ ’ INCOME.tn(h)};

putclose;

Solve negishi_emp minimizing OBJ using emp;

